
TITLE: 
PPLA04: unsteady flow through a deformable perforated plate. 
 
PROBLEM: 
An explosion takes place in a rigid tank containing a deformable perforated plate. The 
plate is submerged below an initial liquid head and is supported along the 
circumference by elastoplastic supports. A cover gas is initially located in the upper 
part of the tank. The calculation is axisymmetric. 
 
MESH: 
The fluid is meshed by FL24 elements, the plate and its support by ED01 conical shell 
elements. Special CL22 boundary condition elements are used to represent the 
coupling between the fluid and the plate. Note that the fluid may freely flow across 
the support, since there is no coupling between them. 
 
MATERIALS: 
FLUT fluid materials are used for the three fluids involved (explosive, liquid and 
cover gas). The deformable structures use an elasto-plastic material law. An 
impedance material (IMPE PPLA) is assigned to the boundary condition elements. 
This represents the pressure drop across the plate, with a given (constant) resistance 
coefficient. 
 
BOUNDARY CONDITIONS: 
The tank is rigid but the inner structures (plate, support) are deformable. Coupling 
between the plate and the fluid is realized by CL22 elements. 

LOADING: 
The event is initiated by the expansion of the high-pressure explosive gas bubble. 
 
CALCULATION: 
The calculation is performed up to a final time of 80 ms. 
 
RESULTS: 
They are analyzed in the reference cited below. 
 
POST-TREATMENT 
An animation is produced. 
 
REFERENCES: 
This problem is detailed, and the results are analyzed, in the following reference: 
1) F. Casadei: “New Boundary Condition Models for Compressible Fluid Flows in 
PLEXIS-3C.” Technical Note N. I.94.75, May 1994. 
 
Numerical Solution 
 
PPLA04 
 
The mesh generation file reads: 
 
*%siz 50 
opti echo 1; 
* 
opti titr 'PPLA - 04'; 
opti dime 2 elem qua4; 
* 
p0=0 0; 
p1=2 0; 
p2=8 0; 
p3=0 7;p3p=p3 plus p0; 
p4=4 7;p4p=p4 plus p0; 
p5=0 9; 
p6=8 9; 
p7=0 12; 
p8=4 12; 
p9=8 12; 
p10=0 2; 
p11=2 2; 
p12=8 2; 
* 
tol=0.01; 
* 
c1=p0 d 2 p1; 
c2=p1 d 2 p11; 
c3=p11 d 2 p10; 
c4=p10 d 2 p0; 
expl=daller c1 c2 c3 c4 plan; 
lag=chan poi1 c2; 
lag=lag et (chan poi1 c3); 
* 
c1=p1 d 6 p2; 
c2=p2 d 2 p12; 
c3=p12 d 6 p11; 
c4=p11 d 2 p1; 
liq1=daller c1 c2 c3 c4 plan; 
* 
c1=p10 d 8 p12; 
c2=p12 d 7 p6; 
c3=p6 d 8 p5; 

c4=p5 d 7 p10; 
liq2=daller c1 c2 c3 c4 plan; 
lag=lag et (chan poi1 c3); 
* 
c1=p5 d 8 p6; 
c2=p6 d 3 p9; 
c3=p9 d 8 p7; 
c4=p7 d 3 p5; 
gas=daller c1 c2 c3 c4 plan; 
* 
liqu=liq1 et liq2; 
flui=expl et liqu et gas; 
* 
clim=p3 d 4 p4; 
* 
elim tol (flui et clim); 
* 
plat=p3p d 4 p4p; 
supp=p4p d 5 p8; 
stru=plat et supp; 
lag=lag et (chan poi1 stru); 
lag=lag et p0 et p2 et p7 et p9; 
* 
mesh=flui et clim et stru; 
* 
blocx=mesh poin droi p0 p7 tol; 
blocx=blocx et (mesh poin droi p2 p9 tol); 
blocy=mesh poin droi p0 p2 tol; 
blocy=blocy et (mesh poin droi p7 p9 tol); 
blocr=p3p et p8; 
* 
mesh=mesh et blocx et blocy et blocr et lag; 
tass mesh; 
* 
opti sauv form 'ppla04.msh'; 
sauv form mesh; 
opti trac ps ftra 'ppla04_mesh.ps'; 
trac qual mesh; 
fin; 

 
The EUROPLEXUS input file reads: 
 
PPLA - 04 
$ 
ECHO 
*CONV win 
AXIS NONL ALE 
CAST MESH 
$ 
DIME 
PT2L 117 PT3L 10 FL24 96 ED01 9 CL22 4 ZONE 3 
NALE 5 BLOQ 48 
 mtpo 2 mtel 2 
TERM 
$ 
GEOM FL24 FLUI ED01 STRU CL22 CLIM TERM 
$ 
COMPL 
EPAI 0.05  LECT PLAT TERM 
     0.10  LECT SUPP TERM 
$ 
GRIL LAGR LECT LAG TERM 
     ALE  LECT FLUI TERM 
     AUTO AUTR 
$ 
MATE VM23 RO 7800. YOUNG 1.6E11 NU 0.333 ELAS 1.05E8 
          TRAC 2 1.05E8 .656256E-3 1.6105E10 1.00066 
          LECT STRU TERM 
     FLUT RO 832. EINT 98.68 GAMM 7.15D0 CL 0.5 
          CQ 2.56 PB 2.71E5 PMIN 0. AHGF 0. ITER 2 
          ALF0 1. BET0 1. KINT 0 NUM 5 
          LECT LIQU TERM 
     FLUT RO 2.4278E3 EINT 0. GAMM 0.75D0 CL 0.5 
          CQ 2.56 PB 0. PMIN 0. AHGF 0. ITER 2 

          ALF0 1. BET0 1. KINT 0 NUM 4 
          LECT EXPL TERM 
     FLUT RO .242777373 EINT 6.865E5 GAMM 1.6 CL 0.5 
          CQ 2.56 PB 0. PMIN 0. AHGF 0. ITER 2 
          ALF0 1. BET0 1. KINT 0 NUM 1 
          LECT  GAS TERM 
     IMPE PPLT ZETA 10.0 
          LECT CLIM TERM 
$ 
LIAI freq 1 
     BLOQ 1 LECT BLOCX TERM 
     BLOQ 2 LECT BLOCY TERM 
     BLOQ 3 LECT BLOCR TERM 
$ 
ECRI VITE ECRO TFRE 40.E-3 
     FICH K200 TFRE 10.E-3 
               POIN TOUS CHAM 
     TRAC TPLO DESC 'PPLA04' TFRE 10.5E-6 
                  POIN LECT 9 118 TERM 
                  ELEM LECT 1 106 TERM 
     fich alic temp FREQ 1 
                    poin lect 9 118 term 
                    elem lect 1 106 term 
     fich alic tfre 1.e-3 
$ 
OPTI NOTEST 
 log 1 
$ 
CALCUL TINI 0. TEND 80.E-3 
* 
SUIT 
Post-treatment 

ECHO 
* 
RESU ALIC TEMP GARD PSCR 
* 
SORT GRAP 
* 
AXTE 1000.0 'Time [ms]'  
* 
COUR 1 'dx_9'   DEPL COMP 1 NOEU LECT 9   TERM 
COUR 2 'dy_9'   DEPL COMP 2 NOEU LECT 9   TERM 
COUR 3 'dy_118' DEPL COMP 2 NOEU LECT 118 TERM 
COUR 4 'p_1'    ECRO COMP 1 ELEM LECT 1   TERM 
COUR 5 'dp_106' ECRO COMP 1 ELEM LECT 106 TERM 
* 

trac 1 2 3 axes 1.0 'DISPL. [M]' 
trac 4     axes 1.0 'P [PA]' 
trac 5     axes 1.0 'DELTAP [PA]' 
* 
QUAL DEPL COMP 1 LECT   9 TERM REFE  2.09508E+0 TOLE 5.E-3 
     DEPL COMP 2 LECT   9 TERM REFE  2.41317E+0 TOLE 5.E-3 
     DEPL COMP 2 LECT 118 TERM REFE  8.67547E-1 TOLE 5.E-3 
     ECRO COMP 1 LECT   1 TERM REFE  1.17908E+6 TOLE 5.E-3 
     ECRO COMP 1 LECT 106 TERM REFE  2.85752E+6 TOLE 5.E-3 
* 
FIN 
 

 
Some results: final fluid pressure and velocity: 
 

 
 
Final structure velocities: 
 

 


